Application Demonstration

The direct search optimization package

Sergey Moiseev 2010

Kodofon

Russia

smoiseev@kodofon.vrn.ru, snmoiseev@yandex.ru

Introduction

The **DirectSearch** package is a collection of commands to numerically computes local and global minimums (maximums) of nonlinear multivariate function with (without) constraints. The package optimization methods are direct searching methods, i.e. they do not require the objective function to be differentiable and continuous.

Package installation instruction

First variant.

1. Find the directory where your Maple installed.
2. Find the subdirectory \lib\
3. Copy the following two files into this subdirectory:
   DirectSearch.mla,
   DirectSearch.hdb.

Second variant.

1. Create the directory, for example, C:\DirectSearchLib\
2. Copy the following two files into this directory:
   DirectSearch.mla,
   DirectSearch.hdb.
3. In current Maple session execute the command:
   > libname:="C:\\DirectSearchLib", libname;

The package command is available via with(DirectSearch) command.
The access to package help is analogues to the other packages help.

```
> restart:
  with(DirectSearch);
```

[GlobalSearch, Search] **(1)**

```
> ?DirectSearch;
```

Local search

The Search finds a minimum of a quadratric function by one iteration because it uses conjugate directions. For quadratric function the checkexit option might be set to 1 without sacrifice of reliability. The solution list contains the final minimum (or maximum) value, the extremum point and a number of objective function evaluations.

```
> f:=(x-2)^2+(y-3)^2+(x-2)*(y-3);
```

$$f := (x-2)^2 + (y-3)^2 + (x-2)\,(y-3) \tag{3.1}$$

```
> Search(f,checkexit=1);
```

$$[0.,\ [x = 2.,\ y = 3.],\ 19] \tag{3.2}$$

Minimize the following quadratic function with n variables. This function has one global minimum (f=0, x=[1000, ..., 1000]).

```
> n:=10;
  f:=unapply(add(add((x[i]-1000)*(x[j]-1000)*exp(-0.0001*abs
  (i-j))*cos(i-j),
  j=1..n),i=1..n),seq(x[i],i=1..n)):
```

$$n := 10 \tag{3.3}$$

```
> Search(f,checkexit=1);
```

$$\left[4.55263667943947\ 10^{-21},\ \left[\begin{array}{c} 1000.00000000119996 \\ 999.99999999945896 \\ 1000.00000000257000 \\ 1000.00000000004002 \\ 999.99999999954298 \\ 1000.00000000358000 \\ 999.99999999808700 \\ 1000.00000000237002 \\ 999.99999999852196 \\ 1000.00000000139004 \end{array}\right],\ 632\right] \tag{3.4}$$

Minimize the following non quadratic function with n variables. This function has one global minimum (f=0, x=[100, ..., 100]).

```
> n:=5;
  f:=add(add((x[i]-100)*(x[j]-100)*exp(-0.01*abs(i-j)),j=1..
  n),i=1..n):
  f:=unapply(1-1/(1+f^2),seq(x[i],i=1..n)):
```

$$n := 5 \tag{3.5}$$

```
> Search(f);
```

$$\left[2.22044604925031\ 10^{-16},\ \left[\begin{array}{c} 100.000001691031912 \\ 99.9995248094204214 \\ 100.000569356193552 \\ 100.000641089629383 \\ 99.9992263476712112 \end{array}\right],\ 349\right] \tag{3.6}$$

Minimize the following function with 3 variables. The initial point is (100,100,100). This function has one global minimum (f=0, x=[1, 0, 0]).

```
> f:=(x,y,z)->100*((z-10*piecewise(x>0,arctan(y/x)/(2*Pi),x=
  0,1,
  1/2+arctan(y/x)/(2*Pi)))^2+(sqrt(x^2+y^2)-1)^2)+z^2;
```

$$f := (x, y, z) \rightarrow 100 \left(\left(z - 10\, piecewise\left(0 < x, \frac{1}{2} \frac{\arctan\left(\frac{y}{x}\right)}{\pi}, x = 0, 1, \frac{1}{2} + \frac{1}{2} \frac{\arctan\left(\frac{y}{x}\right)}{\pi} \right) \right)^2 + 100 \left(\sqrt{x^2 + y^2} - 1 \right)^2 + z^2 \quad (3.7)$$

```
> Search(f,initialpoint=[100,100,100]);
```

$$\left[9.9926684610091\ 10^{-23}, \begin{bmatrix} 0.99999999999904798 \\ 1.17656152787059994\ 10^{-12} \\ 2.09409494950732998\ 10^{-12} \end{bmatrix}, 226 \right] \quad (3.8)$$

Minimize the following function with inequality constraints. This function has one global minimum (f=0, x=[0, 0]).

```
> f:=abs(x-y)+x^2+y^2;
  constr:=[x+y>=0];
```

$$f := |x - y| + x^2 + y^2$$

$$constr := [0 \le x + y] \quad (3.9)$$

```
> Search(f,constr);
```

$$[5.13914663994864\ 10^{-29}, [x = 4.62136324270530542\ 10^{-15}, y = 4.62136324270529674\ 10^{-15}], 47] \quad (3.10)$$

Minimize the following function with inequality and equality constraints. This function has minimum (f=1, x=0, y=1).

```
> f:=x^2+y^2;
  constr:=[x>=0,sqrt(x)+1=ln(y+exp(1)-1)];
```

$$f := x^2 + y^2$$

$$constr := [0 \le x, \sqrt{x} + 1 = \ln(y + e - 1)] \quad (3.11)$$

```
> Search(f,constr,checkexit=4);
```

$$[1.00008441752748, [x = 2.06184438254412990\ 10^{-10}, y = 1.00004220787299003], 3300] \quad (3.12)$$

Global search

Find all minimums of the 4-th Laguerre polynomial. This polynomial has one global and one local minimum.

```
> f:=LaguerreL(4,x);
```

$$f := \mathrm{LaguerreL}(4, x) \quad (4.1)$$

```
> plot(f,x=-2..10);
```




```
> GlobalSearch(f);
```

$$\begin{bmatrix} -9.82294825561955 & [x = 7.75877047449823] & 40 \\ -0.630414938191809 & [x = 0.935822234839441] & 29 \end{bmatrix} \quad \textbf{(4.2)}$$

Find all minimums of the BesselJ function in range 0..50. This function has one global and 7 local minimums.

```
> f:=BesselJ(0,x);
```

$$f := \text{BesselJ}(0, x) \quad \textbf{(4.3)}$$

```
> plot(f,x=0..50);
```



The found solutions are always ranged from the best to the worst

```
> sol:=GlobalSearch(f,[x>=0,x<=50],pointrange=[x=0..50]);
```

$$sol := \begin{bmatrix} -0.402759395702553 & [x = 3.83170599973601] & 18 \\ -0.249704877057843 & [x = 10.1734681743758] & 26 \\ -0.196465371468657 & [x = 16.4706300609734] & 19 \\ -0.167184600473818 & [x = 22.760084415219] & 14 \\ -0.148011109972778 & [x = 29.0468285265161] & 21 \\ -0.134211240310001 & [x = 35.3323075678985] & 21 \\ -0.123667960769837 & [x = 41.6170941215437] & 22 \\ -0.115273694120168 & [x = 47.9014609150159] & 21 \end{bmatrix} \quad (4.4)$$

The global minimum is always the first solution.

```
> sol[1];
```

$$\begin{bmatrix} -0.402759395702553 & [x = 3.83170599973601] & 18 \end{bmatrix} \quad (4.5)$$

Extract the global minimum value and point

```
> sol[1,1];sol[1,2];
```

$$-0.402759395702553$$

$$[x = 3.83170599973601] \quad (4.6)$$

Find the best five minimums of the following function in range x=-3..3, y=-4..4. This function has one global (f=-7.10629975...) and many local minimums.

```
> f:=-3*(1-x)*exp(-x^2-(y+1)^2)+10*(0.2*x-x^3-y^5)*exp(-x^2-y^2)+exp(-(x+1)^2-y^2)/3+cos(x*y);
```

$$f := -3\,(1-x)\,e^{-x^2-(y+1)^2} + 10\,(0.2\,x - x^3 - y^5)\,e^{-x^2-y^2} + \frac{1}{3}\,e^{-(x+1)^2-y^2} + \cos(x\,y) \quad \textbf{(4.7)}$$

```
> plot3d(f,x=-3..3,y=-4..4);
```



```
> GlobalSearch(f,pointrange=[x=-3..3,y=-4..4],solutions=5);
```

$$\begin{bmatrix} -7.10629975630783 & [x=-0.0107231213310914,\ y=1.58135880382854] & 94 \\ -2.51665464519379 & [x=1.27695000295629,\ y=0.0212276648057633] & 68 \\ -1.65709172827052 & [x=-0.297067190313796,\ y=-0.542723718494321] & 76 \\ -1.04678267464566 & [x=2.36548697467322,\ y=-1.28143740642201] & 84 \\ -1.0000222683389 & [x=2.89197484338458,\ y=3.25892837081658] & 186 \end{bmatrix} \quad \textbf{(4.8)}$$

Find all minimums of the following function with constraint. The function has two global minimum (f=2, x=1, x=-1)

```
> f:=x->1/x^2+x^2;
  constr:={x<>0};
```

$$f := x \rightarrow \frac{1}{x^2} + x^2$$

$$constr := \{x \neq 0\} \qquad \textbf{(4.9)}$$

```
> GlobalSearch(f,constr);
```

$$\begin{bmatrix} 2. & [\,1.00000000000147726\,] & 30 \\ 2. & [\,-1.00000000296385227\,] & 31 \end{bmatrix} \qquad \textbf{(4.10)}$$

Find the best two minimums of the following function in range x=-8..1, y=-3..10. This function has one global (f=0,x=0,y=0) and many local minimums.

```
> f:=(x,y)->x^2+y^2+100*sin(x^2+x+y^2-y)^2;
  constr:=[x>=-8,x<=1,y>=-3,y<=10];
```

$$f := (x, y) \rightarrow x^2 + y^2 + 100 \sin\left(x^2 + x + y^2 - y\right)^2$$

$$constr := [-8 \leq x, x \leq 1, -3 \leq y, y \leq 10] \qquad \textbf{(4.11)}$$

```
> plot3d(f(x,y),x=-8..1,y=-3..10,numpoints=5000);
```



```
> GlobalSearch(f,constr,pointrange=[x=-8..1,y=-3..10],
  solutions=2);
```

$$\begin{bmatrix} 3.90811532489163\ 10^{-19} & \begin{bmatrix} 2.43976117127799994\ 10^{-10} \\ 2.93488889227798986\ 10^{-10} \end{bmatrix} & 94 \\ 1.4418646391686 & \begin{bmatrix} 0.848785779219560998 \\ -0.848785777504651006 \end{bmatrix} & 131 \end{bmatrix} \qquad \textbf{(4.12)}$$

Find global maximum of the following function in range x=-1..3, y=-2..3 with given constraints. This function has one global maximum(f=3.60979...) and some local maximums.

```
> f:=(x,y)->(2*sin(1+2*x)*sin(3*x+y))^2;
  constr:=[x>=-1,x<=3,y>=-2,y<=3,x^2-3*x*y>=0,y*sin(y)-sin(x)
  <=0.3,(x-y)^2<=0.1];
```

$$f := (x, y) \rightarrow 4\sin(1+2x)^2 \sin(3x+y)^2$$

$$constr := [-1 \leq x, x \leq 3, -2 \leq y, y \leq 3, 0 \leq x^2 - 3xy, y\sin(y) - \sin(x) \leq 0.3, (x - y)^2 \leq 0.1] \qquad \textbf{(4.13)}$$

```
> plot3d(f(x,y),x=-1..3,y=-2..3,numpoints=5000);
```



```
> GlobalSearch(f,constr,pointrange=[x=-1..3,y=-2..3],
  solutions=1,maximize);
```

(4.14)

$$\left[\begin{array}{ccc} 3.60979413598907 & \left[\begin{array}{c} 0.421469255450298341 \\ 0.140489751805238872 \end{array}\right] & 66 \end{array}\right] \tag{4.14}$$

For more examples see package Help